P
L
/
S
Q
L

Oracle11*g*:
PL/SQL Programming

Chapter 4

Cursors and Exception Handling

Chapter Objectives

PL/SQL

- After completing this lesson, you should be able to understand:
  - Manipulating data with cursors
  - Using bulk-processing features
  - Managing errors with exception handlers
  - Addressing exception-handling issues, such as RAISE_APPLICATION_ERROR and propagation
  - Documenting code with comments

Brewbean's Challenge

P L / S Q L

- Processing multiple data rows



Cursors

PL/SQL

- Work area in which SQL statement is processed
- Implicit cursor – declared automatically for DML and SELECT statements
- Explicit cursor – declared and managed programmatically to handle a set of rows returned by a SELECT statement
- Cursor variable – reference or pointer to a work area or cursor

Cursor Attributes

P L / S Q L

Attribute Name	Data type	Description
%ROWCOUNT	Number	Number of rows affected by the SQL statement
%FOUND	Boolean	TRUE if at least one row is affected by the SQL statement, otherwise FALSE
%NOTFOUND	Boolean	TRUE if no rows are affected by the SQL statement, otherwise FALSE

Implicit Cursor

P
L
/
S
Q
L



Explicit Cursor

PL/SQL

Step	Step Activity	Activity Description
1	DECLARE	Creates a named cursor identified by a SELECT statement. The SELECT statement does not include an INTO clause. Values in the cursor are moved to PL/SQL variables with the FETCH step.
2	OPEN	Processes the query and creates the active set of rows available in the cursor.
3	FETCH	Retrieves a row from the cursor into block variables. Each consecutive FETCH issued will retrieve the next row in the cursor until all rows have been retrieved.
4	CLOSE	Clears the active set of rows and frees the memory area used for the cursor.

Explicit Cursor Example

P L / S Q L

```
DECLARE
   CURSOR cur_basket IS
     SELECT bi.idBasket, p.type, bi.price, bi.quantity
       FROM bb_basketitem bi INNER JOIN bb_product p
         USING (idProduct)
       WHERE bi.idBasket = :g_basket;
   TYPE type_basket IS RECORD (
     basket bb_basketitem.idBasket%TYPE,
     type bb_product.type%TYPE,
     price bb_basketitem.price%TYPE,
     qty bb_basketitem.quantity%TYPE );
   rec_basket type_basket;
   lv_rate_num NUMBER(2,2);
   lv_tax_num NUMBER(4,2) := 0;
BEGIN
   OPEN cur_basket;
   LOOP
     FETCH cur_basket INTO rec_basket;
      EXIT WHEN cur_basket%NOTFOUND;
      IF rec_basket.type = 'E' THEN lv_rate_num := .05; END IF;
      IF rec_basket.type = 'C' THEN lv_rate_num := .03; END IF;
      lv_tax_num := lv_tax_num + ((rec_basket.price*rec_basket.qty)*lv_rate_num);
   END LOOP;
   CLOSE cur_basket;
   DBMS_OUTPUT.PUT_LINE(lv_tax_num);
END;
/
```

Declare cursor

Declare record type and variable

Open cursor

Fetch a row from the cursor

Check if row returned from fetch

Close cursor

Calculate tax amount

Cursor FOR Loop

P L / S Q L

- Handles tasks automatically for processing each row returned by a cursor (record declaration, fetch, ending loop)
- Use FOR UPDATE and WHERE CURRENT OF clauses for record locking

Cursor FOR Loop Example

P
L
/
S
Q
L

```
DECLARE
  CURSOR cur_prod IS
       SELECT type, price
         FROM bb_product
         WHERE active = 1
       FOR UPDATE NOWAIT;
  lv_sale bb_product.saleprice%TYPE;
BEGIN
 FOR rec_prod IN cur_prod LOOP
    IF rec_prod.type = 'C' THEN lv_sale := rec_prod.price * .9;
     ELSIF rec_prod.type = 'E' THEN lv_sale := rec_prod.price * .95;
    END IF;
    UPDATE bb_product
      SET saleprice = lv_sale
      WHERE CURRENT OF cur_prod;
  END LOOP;
COMMIT;
END;
```

Cursors with Parameters

PL/SQL

- Use parameters to make dynamic
- Parameters are values passed to the cursor when it is opened
- Enables the cursor to retrieve different data based on the input values

Cursors with Parameters

P L / S Q L

```
DECLARE
    CURSOR cur_order (p_basket NUMBER) IS
      SELECT idBasket, idProduct, price, quantity
       FROM bb_basketitem
       WHERE idBasket = p_basket;
    lv_bask1_num bb_basket.idbasket%TYPE := 6;
    lv_bask2_num bb_basket.idbasket%TYPE := 10;
BEGIN
  FOR rec_order IN cur_order(lv_bask1_num) LOOP
     DBMS_OUTPUT.PUT_LINE(rec_order.idBasket || ' - ' ||
                    rec_order.idProduct || ' - ' || rec_order.price);
  END LOOP;
  FOR rec_order IN cur_order(lv_bask2_num) LOOP
     DBMS_OUTPUT.PUT_LINE(rec_order.idBasket || ' - ' ||
                    rec_order.idProduct || ' - ' || rec_order.price);
  END LOOP;
END;
```

Cursor Variable

PL/SQL

- More efficiently handles data returned by query by returning a pointer to the work area rather than the actual result set
- The same cursor variable can be used for different query statements

Cursor Variable Example

P L / S Q L

```
DECLARE
      cv_prod SYS_REFCURSOR;
      rec_item bb_basketitem%ROWTYPE;
      rec_status bb_basketstatus%ROWTYPE;
      lv_input1_num NUMBER(2) := 2;
      lv_input2_num NUMBER(2) := 3;
BEGIN
      IF lv_input1_num = 1 THEN
         OPEN cv_prod FOR SELECT * FROM bb_basketitem
           WHERE idBasket = lv_input2_num;
          LOOP
            FETCH cv_prod INTO rec_item;
            EXIT WHEN cv_prod%NOTFOUND;
            DBMS_OUTPUT.PUT_LINE(rec_item.idProduct);
          END LOOP;
```

Example (continued)

P L / S Q L

```
ELSIF lv_input1_num = 2 THEN
    OPEN cv_prod FOR SELECT * FROM bb_basketstatus
                                WHERE idBasket = lv_input2_num;
        LOOP
            FETCH cv_prod INTO rec_status;
            EXIT WHEN cv_prod%NOTFOUND;
            DBMS_OUTPUT.PUT_LINE(rec_status.idStage || ' - '
                                                || rec_status.dtstage);
        END LOOP;
 END IF;
END;
```

Bulk-processing

P L / S Q L

- Improve performance of multirow queries and DML statements
- Processes groups of rows without context switching between the SQL and PL/SQL processing engine
- Use in FETCH with LIMIT clause
- FORALL option with DML activity

Bulk-processing (Query)

P L / S Q L

```
DECLARE
     CURSOR cur_item IS
          SELECT *
          FROM bb_basketitem;
     TYPE type_item IS TABLE OF cur_item%ROWTYPE
                         INDEX BY PLS_INTEGER;
      tbl_item type_item;
BEGIN
      OPEN cur_item;
      LOOP
          FETCH cur_item BULK COLLECT INTO tbl_item LIMIT 1000;
          FOR i IN 1..tbl_item.COUNT LOOP
            DBMS_OUTPUT.PUT_LINE(tbl_item(i).idBasketitem || ' -'
                                                    || tbl_item(i).idProduct);
          END LOOP;
          EXIT WHEN cur_item%NOTFOUND;
      END LOOP;
      CLOSE cur_item;
END;
```

Bulk-processing (DML)

P L / S Q L

```
DECLARE
    TYPE emp_type IS TABLE OF NUMBER INDEX
          BY BINARY_INTEGER;
    emp_tbl emp_type;
BEGIN
     SELECT empID
       BULK COLLECT INTO emp_tbl
       FROM employees
         WHERE classtype = '100';
     FORALL i IN d_emp_tbl.FIRST .. emp_tbl.LAST
       UPDATE employees
            SET raise = salary * .06
            WHERE empID = emp_tbl(i);
       COMMIT;
END;
```

Exception Handlers

PL/SQL

- Used to capture error conditions and handle the processing to allow the application to continue
- Placed in the EXCEPTION section of a PL/SQL block
- Two types of errors
  1. Oracle errors (Predefined and Non-Predefined)
  2. User-defined errors
- RAISE_APPLICATION_ERROR

Predefined Oracle Errors

Exception Name	Description
NO_DATA_FOUND	A SELECT statement in a PL/SQL block retrieves no rows or a nonexistent row of an index-by table is referenced
TOO_MANY_ROWS	A SELECT statement in a PL/SQL block retrieves more than one row
CASE_NOT_FOUND	No WHEN clause in the CASE statement is processed
ZERO_DIVIDE	Attempted division by zero
DUP_VAL_ON_INDEX	Attempted violation of a unique or primary key column constraint

Predefined Error Example

P L / S Q L



Undefined Error

P
L
/
S
Q
L

- Identify possible errors for statements in a block



Handler Added

P
L
/
S
Q
L



User-Defined Exception

PL/SQL

- No system error is raised
- Raise errors to enforce business rules
- Once error is raised, the remaining statements in the executable sections are not executed
- Processing moves to the exception area of the block

User-Defined Exception Example

P L / S Q L

XE_plbook

0.046 seconds

XE_plbook

Worksheet | Query Builder

```
1  DECLARE
2    ex_prod_update EXCEPTION;
3  BEGIN
4   UPDATE bb_product
5    SET description = 'Mill grinder with 5 grind settings!'
6    WHERE idProduct = 30;
7   IF SQL%NOTFOUND THEN
8     RAISE ex_prod_update;
9   END IF;
10 EXCEPTION
11   WHEN ex_prod_update THEN
12     DBMS_OUTPUT.PUT_LINE('Invalid product ID entered');
13 END;
```

Declare an exception name

If no rows are updated, raise the exception

Establish an exception handler

Script Output

Task completed in 0.046 seconds

```
anonymous block completed
```

Dbms Output

Buffer Size: 20000

```
Invalid product ID entered
```

XE_plbook

User-Defined Exception Example

P L / S Q L

XE_plbook

0 seconds

XE_plbook

Worksheet | Query Builder

```
1  DECLARE
2    lv_ordqty_num NUMBER(2) := 20;
3    lv_stock_num bb_product.stock%TYPE;
4    ex_prod_stk EXCEPTION;
5  BEGIN
6   SELECT stock
7    INTO lv_stock_num
8    FROM bb_product
9    WHERE idProduct = 2;
10  IF lv_ordqty_num > lv_stock_num THEN
11    RAISE ex_prod_stk;
12  END IF;
13 EXCEPTION
14   WHEN ex_prod_stk THEN
15     DBMS_OUTPUT.PUT_LINE('Requested quantity beyond stock level');
16     DBMS_OUTPUT.PUT_LINE('Req qty = ' || lv_ordqty_num ||
17                          'Stock qty = ' || lv_stock_num);
18 END;
```

Declare an exception name

If the quantity requested is greater than the quantity in stock, raise the exception

Establish an exception handler

Script Output

Task completed in 0 seconds

```
anonymous block completed
```

Dbms Output

Buffer Size: 20000

```
Requested quantity beyond stock level
Req qty = 20Stock qty = 15
```

Additional Exception Concepts

P L / S Q L

- WHEN OTHERS – traps all errors not specifically addressed by an exception handler and used for handling unanticipated errors
- SQLCODE and SQLERRM – functions used to identify the error code and message, especially in application, testing to identify unanticipated errors

Example

P L / S Q L

XE_plbook

0.016 seconds

XE_plbook

Worksheet | Query Builder

```
24  EXCEPTION
25    WHEN NO_DATA_FOUND THEN
26      DBMS_OUTPUT.PUT_LINE('You have no saved baskets!');
27    WHEN OTHERS THEN
28      lv_errmsg_txt := SUBSTR(SQLERRM,1,80);
29      lv_errnum_txt := SQLCODE;
30      INSERT INTO bb_trans_log (shopper, appaction, errcode, errmsg)
31          VALUES(lv_shopper_num, 'Get saved basket',lv_errnum_txt, lv_errmsg_txt
32      DBMS_OUTPUT.PUT_LINE('A problem has occurred');
33      DBMS_OUTPUT.PUT_LINE('Tech support will be notified and contact you');
34  END;
```

Script Output

Task completed in 0.016 seconds

```
anonymous block completed
```

Dbms Output

Buffer Size: 20000

```
A problem has occurred
Tech support will be notified and contact you
```

XE_plbook

Exception Propagation

P L / S Q L

- Exception handling in nested blocks
- Exception raised in a block will first look for handler in the exception section of that block, if no handler found, execution will move to the exception section of the enclosing block
- Error in DECLARE section propagates directly to exception section of the enclosing block
- Error in exception handler propagates to exception section of the enclosing block

Exception Propagation

P L / S Q L

XE_plbook

0.016 seconds

XE_plbook

Worksheet | Query Builder

```
13     EXCEPTION
14       WHEN NO_DATA_FOUND THEN
15           DBMS_OUTPUT.PUT_LINE('No data error in nested block');
16     END;
17     lv_junk_num := 3;
18   EXCEPTION
19    WHEN OTHERS THEN
20     DBMS_OUTPUT.PUT_LINE('Error Code = '||SQLCODE);
21     DBMS_OUTPUT.PUT_LINE('Error Message = '||SQLERRM);
22   END;
```

Script Output

Task completed in 0.016 seconds

```
anonymous block completed
```

Dbms Output

Buffer Size: 20000

```
Error Code = -1422
Error Message = ORA-01422: exact fetch returns more than requested number of rows
```

XE_plbook

Commenting Code

P L / S Q L

- Add comments within code to identify code purpose and processing steps
- Use /* */ to enclose a multiline comment
- Use -- to add a single or partial line comment

Comment Examples

P L / S Q L

```
DECLARE
   ex_prod_update EXCEPTION;  --For UPDATE of no rows
   exception
BEGIN
 /* This block is used to update product descriptions
    Constructed to support the Prod_desc.frm app screen
      Exception raised if no rows updated  */
   UPDATE bb_product
    SET description = 'Mill grinder with 5 grind settings!'
    WHERE idProduct = 30;
  --Check if any rows updated
 IF SQL%NOTFOUND THEN
    RAISE ex_prod_update;
   END IF;
EXCEPTION
   WHEN ex_prod_update THEN
     DBMS_OUTPUT.PUT_LINE('Invalid product id entered');
END;
```

Summary

P L / S Q L

- Implicit cursors are automatically created for SQL statements
- Explicit cursors are declared
- Cursors allow the processing of a group of rows
- CURSOR FOR Loops simplify cursor coding
- Parameters make cursors more dynamic
- A REF CURSOR acts like a pointer
- BULK processing options can improve performance for queries and DML activity

Summary (continued)

P L / S Q L

- Add error handlers in the EXCEPTION area to manage Oracle and user-defined errors
- Exception propagation is the flow of error handling processing
- Use comments in code for documentation